Dpto. de Ciencias de la Computación e Inteligencia Artificial UNIVERSIDAD DE SEVILLA Dpto. de Lenguajes y Sistemas Informáticos UNIVERSIDAD DE CADIZ

Automatic Verification of

Polynomial Rings Fundamental Properties in ACL2

Inmaculada Medina Bulo et al.

ACL2 Workshop 2000.

Page 1 Introduction

Introduction

▶ Goals

– Formalization of multivariate polynomials over a coefficient field, Q, and their basic operations in ACL2 – Verification of their main properties – Computation by using their operations

▶ Main findings

– Polynomial formalization – Automatic verification of fundamental properties that structure them as a ring

∗ ⟨K[X], +, -, 0⟩ form a commutative group ∗ ⟨K[X], ·, 1⟩ form a commutative monoid ∗ · is distributive over + on the right and on the left

– Computation by using the operations

▶ Potential applications

– The formalization of Buchberger’s algorithm

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 2 Polynomial Representation Problems

Polynomial Representation Problems

▶ Normalized/Unnormalized Representation

1. Normalized Representation and Syntactic Equality – Advantages Equality is syntactic and ACL2 handles it directly (EQUAL) – Disadvantages We have to work in normal form. This complicates the proofs 2. Unnormalized Representation and Semantic Equal- ity – Advantages It spares the operations from the need to work with normal forms. The computation done by the algorithm is separated from the normalization process – Disadvantages Equality must work module normal form and the prover does not manage it directly

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 3 Polynomial Representation Problems

Polynomial Representation Problems

▶ Dense/Sparse Representation

1. Dense – Advantages Simple algorithms – Disadvantages Unsuitable for the case of multiple variables 2. Sparse – Advantages Suitable for multivariate polynomials – Disadvantages More complex algorithms

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 4 Polynomial Representation

Polynomial Representation

We have chosen

▶ Initially, a sparse normalized representation ▶ Finally, a sparse unnormalized representation ▶ Formalization

– A polynomial is a finite sum of monomials – A semantic equality predicate – Necessary operations (addition, negation, multiplication) – Verification of the fundamental properties of polynomials – A monomial is a product between a coefficient and a term

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 5 Formalization of Terms

Formalization of Terms

▶ Definition

A term on a set of variables X is a finite power product of the form x^{e1}_1 . . . x^{en}_n = X^{⟨e1,...,en⟩} = n∏ i=1 x^{ei}_i ▶ Representation

A list of natural numbers, once we have determined X and an order <X over X.

For example, <X= f(xi, xj) : i < jg x^{e1}_1 · · · · · x^{en}_n = X^{⟨e1,...,en⟩} −) ⟨e_1, . . . , e_n⟩ ▶ Null Term 1 = x^0_1 · · · · · x^0_n = X^{⟨0,...,0⟩} −) ⟨0, . . . , 0⟩

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 6 Terms in ACL2

Terms in ACL2

▶ Recognizer of terms

(defmacro termp (a)
  ‘(natural-listp ,a))

▶ Null term

(defconst *null*
  nil)

(defun nullp (a)
  (cond ((atom a)
         (equal a *null*))
        (t
         (and (equal (first a) 0)
              (nullp (rest a))))))

▶ Compatibility and equality relation

(defmacro compatiblep (a b)
  ‘(equal (len ,a) (len ,b)))

(defmacro = (a b)
  ‘(equal ,a ,b))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 7 Multiplication of Terms

Multiplication of Terms

▶ Definition X^{⟨a1,...,an⟩} · X^{⟨b1,...,bn⟩} = X^{⟨a1+b1,...,an+bn⟩}

(defun * (a b)
  (cond ((and (not (termp a)) (not (termp b)))
         *null*)
        ((not (termp a)) b)
        ((not (termp b)) a)
        ((endp a) b)
        ((endp b) a)
        (t
         (cons (LISP::+ (first a) (first b))
               (* (rest a) (rest b))))))

▶ Commutative Monoid Structure

(defthm *-identity-1
  (implies (and (nullp a) (termp b)
                (compatiblep a b))
           (= (* a b) b)))

(defthm *-identity-2
  (implies (and (termp a) (nullp b)
                (compatiblep a b))
           (= (* a b) a)))

(defthm associativity-of-*
  (= (* (* a b) c) (* a (* b c))))

(defthm commutativity-of-*
  (= (* a b) (* b a)))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 8 Total and Strict Order on Terms

Total and Strict Order on Terms

▶ Definition (lexicographical ordering) X^{⟨a1,...,an⟩} < X^{⟨b1,...,bn⟩} ≡ ⟨a_1, . . . , a_n⟩ < ⟨b_1, . . . , b_n⟩ ≡ ∃i (a_i < b_i ^ ∀j < i a_j = b_j)

(defun < (a b)
  (cond ((or (endp a) (endp b))
         (not (endp b)))
        ((equal (first a) (first b))
         (< (rest a) (rest b)))
        (t
         (LISP::< (first a) (first b)))))

▶ Properties of the order

(defthm irreflexivity-of-<
  (not (< a a)))

(defthm transitivity-of-<
  (implies (and (< a b) (< b c))
           (< a c)))

(defthm trichotomy-of-<
  (implies (and (termp a) (termp b))
           (or (< a b) (< b a) (= a b)))
  :rule-classes nil)

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 9 Term Embedding in ε0-ordinals

Term Embedding in ε0-ordinals

▶ Formalization X^{⟨a1,...,an⟩} −) ω^{ωn+a1} + · · · + ω^{ω+an} x

(1)

−) ω^{ω+1}

((1 . 1) . 0)

x^8 · y^0

(8 0)

−) ω^{ω2+8} + ω^ω

((2 . 8) (1 . 0) . 0)

x^4 · y^3 · z^5

(4 3 5)

−) ω^{ω3+4} + ω^{ω2+3} + ω^{ω+5}

((3 . 4) (2 . 3) (1 . 5) . 0)

▶ Definition

(defun term->e0-ordinal (a)
  (cond ((endp a)
         0)
        (t
         (cons (cons (len a) (first a))
               (term->e0-ordinal (rest a))))))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 10 Well-founded Order

Well-founded Order

(defthm e0-ordinalp-term->e0-ordinal
  (implies (termp a)
           (e0-ordinalp (term->e0-ordinal a)))
  :hints (("Goal" ...)))

(defthm well-ordering-of-<
  (and (implies (termp a)
                (e0-ordinalp
                   (term->e0-ordinal a)))
       (implies (and (termp a) (termp b)
                     (< a b))
             (e0-ord-< (term->e0-ordinal a)
                       (term->e0-ordinal b))))
  :rule-classes :well-founded-relation)

▶ Problem

(< ’(3 1) ’(1 2 1)) −) nil

(term->e0-ordinal ’(3 1))

−) ((2 . 3) (1 . 1) . 0)

(term->e0-ordinal ’(1 2 1))

−) ((3 . 1) (2 . 2) (1 . 1) . 0)

(e0-ord-< ’((2 . 3) (1 . 1) . 0)
          ’((3 . 1) (2 . 2) (1 . 1) . 0))

−) t ▶ Solution

(defun < (a b)
  (cond ((LISP::< (len a) (len b))
         t)
        ((LISP::> (len a) (len b))
         nil)
        (...

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 11 Admissibility of the Order

Admissibility of the Order

▶ Definition – ∀a ∈ [X]nf1g 1 = x^0_1 · · · · · x^0_n < a – ∀a, b, c ∈ [X] (a < b =} ac < bc) ▶ Formalization

– The order has a first element

(defthm <-has-first
  (implies (and (termp a) (termp b)
                (compatiblep a b)
                (nullp a) (not (nullp b)))
           (< a b)))

– The order is compatible with the multiplication

(defthm <-compatible-*-1
  (implies (and (termp a) (termp b)
                (termp c)
                (compatiblep a c)
                (compatiblep b c)
                (< a b))
           (< (* a c) (* b c))))

(defthm <-compatible-*-2
  (implies (and (termp a) (termp b)
                (termp c)
                (compatiblep a c)
                (compatiblep b c)
                (< a b))
           (< (* c a) (* c b))))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 12 Formalization of Monomials

Formalization of Monomials

▶ Definition

A monomial on X is a product of the form c · X^{⟨e1,...,en⟩} ▶ Representation

A list whose first element is its coefficient and whose rest is its term c · X^{⟨e1,...,en⟩} −) (c (e_1 . . . e_n)) ▶ Identity Monomial 1 · X^{⟨0,...,0⟩} −) (1 (0 . . . 0))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 13 Monomials in ACL2

Monomials in ACL2

▶ Recognizer of monomials

(defmacro monomialp (a)
  ‘(and (consp ,a) (rationalp (first ,a))
        (termp (rest ,a))))

▶ Identity and Null Monomial

(defconst *one* (monomial 1 TER::*null*))

(defmacro onep (a)
  ‘(and (equal (coefficient ,a) 1)
        (TER::nullp (term ,a))))

(defconst *null* (monomial 0 TER::*null*))

(defmacro nullp (a)
  ‘(equal (coefficient ,a) 0))

▶ Compatibility and equality relation

(defun compatiblep (a b)
  (TER::compatiblep (term a) (term b)))

(defun = (a b)
  (or (and (nullp a) (nullp b))
      (and (LISP::= (coefficient a)
                    (coefficient b))
           (TER::= (term a) (term b)))))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 14 Multiplication of Monomials

Multiplication of Monomials

▶ Definition aX^{⟨a1,...,an⟩} · bX^{⟨b1,...,bn⟩} = (a · b)X^{⟨a1+b1,...,an+bn⟩}

(defun * (a b)
  (monomial (LISP::* (coefficient a)
                     (coefficient b))
            (TER::* (term a) (term b))))

▶ Commutative Monoid Structure

(defthm *-identity-1
  (implies (and (onep a) (monomialp b)
                (compatiblep a b))
           (= (* a b) b)))

(defthm *-identity-2
  (implies (and (monomialp a) (onep b)
                (compatiblep a b))
           (= (* a b) a)))

(defthm associativity-of-*
  (= (* (* a b) c) (* a (* b c))))

(defthm commutativity-of-*
  (= (* a b) (* b a)))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 15 Formalization of Polynomials

Formalization of Polynomials

▶ Definition

A polynomial on X is a finite sum of monomials c_1 · X^{⟨e11,...,e1n⟩} + · · · + c_m · X^{⟨em1,...,emn⟩} ▶ Representation ((c_1 (e_{11} . . . e_{1n})) . . . (c_m (e_{m1} . . . e_{mn}))) ▶ Recognizer

(defmacro polynomialp (p)
  ‘(monomial-listp ,p))

▶ Null polynomial

(defconst *null* nil)
(defmacro nullp (p) ‘(endp ,p))

▶ Identity polynomial

(defconst *one*
  (polynomial MON::*one* *null*))
(defmacro onep (p) ‘(= ,p *one*))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 16 Polynomial Semantic Equality

Polynomial Semantic Equality

The equality relation defined on polynomials must verify the following properties 1. The reflexive property 2. The symmetrical property 3. The transitive property

4. p_1 +_p (m +_m p_2) =_p m +_m (p_1 +_p p_2) 5. p_1 =_p p_2 ^ q_1 =_p q_2 =} p_1 +_p q_1 =_p p_2 +_p q_2 6. (k_1, t)+_m ((k_2, t)+_m p) =_p ((k_1 +_k k_2), t)+_m p 7. m = 0 =} m +_m p =_p p

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 17 Polynomial Equality in ACL2

Polynomial Equality in ACL2

▶ Semantic equality

(defun = (p1 p2)
  (equal (nf p1) (nf p2)))

▶ Normal form

– Monomials are strictly ordered

– Null monomials do not appear

This implies that monomials with identical terms can

not appear

(defun nf (p)
  (cond ((or (not (polynomialp p)) (nullp p))
         *null*)
      (t
       (+-monomial (first p) (nf (rest p))))))

(defun +-monomial (m p)
  (cond
   ((MON::nullp m) p)
   ((nullp p) (polynomial m *null*))
   ((TER::= (term m) (term (first p)))
    (let ((c (LISP::+ (coefficient m)
                    (coefficient (first p)))))
      (if (equal c 0) (rest p)
        (polynomial (monomial c (term m))
                    (rest p)))))
   ((TER::< (term (first p)) (term m))
    (polynomial m p))
   (t
    (polynomial (first p)
                (+-monomial m (rest p))))))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 18 Addition & Negation of Polynomials

Addition & Negation of Polynomials

▶ Polynomial Addition

(defun + (p1 p2)
  (cond ((and (not (polynomialp p1))
              (not (polynomialp p2))) *null*)
        ((not (polynomialp p1)) p2)
        ((not (polynomialp p2)) p1)
        (t (append p1 p2))))

▶ Polynomial Negation

(defun - (p)
  (cond ((or (not (polynomialp p)) (nullp p))
         *null*)
        (t (polynomial
            (monomial (LISP::- (coefficient
                                   (first p)))
                      (term (first p)))
            (- (rest p))))))

▶ Commutative Group with Addition and Negation

(defthm --distributes-+
  (= (- (+ p1 p2)) (+ (- p1) (- p2))))

(defthm +-identity-1 (= (+ p *null*) p))

(defthm +-identity-2 (= (+ *null* p) p))

(defthm associativity-of-+
  (= (+ (+ p1 p2) p3) (+ p1 (+ p2 p3))))

(defthm commutativity-of-+
  (= (+ p1 p2) (+ p2 p1)))

(defthm +-- (= (+ p (- p)) *null*)))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 19 Multiplication of Polynomials

Multiplication of Polynomials

▶ Definition

(defun *-monomial (m p)
  (cond ((or (nullp p) (not (monomialp m))
             (not (polynomialp p)))
         *null*)
      (t (polynomial (MON::* m (first p))
                   (*-monomial m (rest p))))))

(defun * (p1 p2)
  (cond ((or (nullp p1)
             (not (polynomialp p1)))
         *null*)
        (t
         (+ (*-monomial (first p1) p2)
            (* (rest p1) p2)))))

▶ Commutative Monoid with Multiplication

(defthm *-identity-1 (= (* *one* p) p))

(defthm *-identity-2 (= (* p *one*) p))

(defthm associativity-of-*
  (= (* p1 (* p2 p3)) (* (* p1 p2) p3)))

(defthm commutativity-of-*
  (= (* p1 p2) (* p2 p1)))

(defthm *-cancellative-1
  (= (* *null* p) *null*))

(defthm *-cancellative-2
  (= (* p *null*) *null*))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 20 Distributivity Property

Distributivity Property

▶ Distributivity of Multiplication over Addition

(defthm *-distributes-+-1
  (= (* p1 (+ p2 p3))
     (+ (* p1 p2) (* p1 p3))))

(defthm *-distributes-+-2
  (= (* (+ p1 p2) p3)
     (+ (* p1 p3) (* p2 p3))))

This completes the proof that polynomials have a ring

structure

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 21 Congruences

Congruences

▶ Polynomial constructor

(defcong MON::= = (polynomial m p) 1)

(defcong = = (polynomial m p) 2)

▶ Negation and addition of polynomials

(defcong = = (- p) 1))

(defcong = = (+ p1 p2) 2)
(defcong = = (+ p1 p2) 1)

▶ Multiplication between monomials and polynomials

(defcong MON::= = (*-monomial m p) 1)

(defthm =-implies-=-*-monomial-2
 (implies (and (monomialp m)
               (polynomialp p1)
               (polynomialp p1-equiv)
               (MON::compatiblep m (first p1))
               (compatiblep p1 p1-equiv)
               (= p1 p1-equiv))
          (= (*-monomial m p1)
             (*-monomial m p1-equiv))))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 22 Congruences

Congruences

▶ Multiplication of polynomials

(defthm =-implies-=-*-2
  (implies (and (polynomialp p1)
                (polynomialp p2)
                (polynomialp p2-equiv)
                (compatiblep p1 p2)
                (compatiblep p1 p2-equiv)
                (= p2 p2-equiv))
           (= (* p1 p2) (* p1 p2-equiv))))

(defthm =-implies-=-*-1
  (implies (and (polynomialp p1)
                (polynomialp p1-equiv)
                (polynomialp p2)
                (compatiblep p1 p2)
                (compatiblep p1-equiv p2)
                (= p1 p1-equiv))
           (= (* p1 p2) (* p1-equiv p2))))

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.

Page 23 Conclusions and Future Work

Conclusions and Future Work

▶ Conclusions

– A formalization of multivariate polynomials rings with rational coefficients in ACL2 – It is interesting to note some of the advantages exposed by ACL2 in comparison with NQTHM – Compatibility relation complicate the proofs – Guards: Operations can be executed on any platform

▶ Future Work

– Abstraction of the coefficient field – Obtaining an automatic verification of Buchberger’s algorithm for Gröbner bases – There are many applications of Gröbner bases

Inmaculada Medina Bulo et al. ACL2 Workshop 2000.